CSc 520

Principles of Programming Languages

51: Semantics — Syntax

Christian Collberg
collberg@cs.arizona.edu

Department of Computer Science
University of Arizona

Copyright © 2005 Christian Collberg
Syntax

The syntax of a language (formal or natural) is the way the words in a sentence/program can be arranged.

```
eats dog bone the
```

is not a legal arrangement of words in English.

```
= y x + 5
```

is not a legal arrangement of tokens in Java.

Somehow, we need to describe what constitutes legal and illegal sentences in a particular language.

We use production rules to describe the syntax of a language.
Here's a production rule:

\[ \text{IfStat} \rightarrow \text{if} \ ( \text{expr} ) \ \text{stat} \]

This rule states that to construct an if-statement in C you have to type
1. an \text{if}, then
2. a (, then
3. some sort of expression, then
4. a ), then finally
5. some sort of statement.
A grammar can be used for
1. sentence generation (i.e. which sentences does this grammar generate?), or
2. parsing (i.e. is sentence $S$ generated by this grammar?).

Let’s look at a simple grammar for a fragment of English.
Syntactic Categories

S  [Sentence] John likes Sarah’s black hair
N  [Noun] John, hair
V  [Verb] eating, sat
Adj [Adjective] black, long
Det [Determiner] the, a, every
NP [Noun Phrase] Sarah’s long black hair
VP [Verb Phrase] eating apples
# A Simple English Grammar

<table>
<thead>
<tr>
<th>S</th>
<th>→</th>
<th>NP VP</th>
</tr>
</thead>
<tbody>
<tr>
<td>VP</td>
<td>→</td>
<td>V NP</td>
</tr>
<tr>
<td>VP</td>
<td>→</td>
<td>V</td>
</tr>
<tr>
<td>NP</td>
<td>→</td>
<td>N</td>
</tr>
<tr>
<td>NP</td>
<td>→</td>
<td>Det N</td>
</tr>
</tbody>
</table>

| N     | → | John |
| N     | → | Lisa |
| N     | → | house|
| V     | → | died |
| V     | → | kissed|
| Det   | → | the  |
| Det   | → | a    |

- S, NP, VP, N, Det, V are **non-terminal** symbols.
- John, Lisa, house, died, ... are **terminal symbols**.
- S is the **start symbol**.
Sentence Generation

1. Start with the start symbol.
2. Pick a non-terminal $X$ on the right hand side.
3. Pick a grammar rule $X \rightarrow \gamma$.
4. Replace $X$ with $\gamma$.
5. Repeat until left with a string of words.

<table>
<thead>
<tr>
<th>Rule</th>
<th>Replacement</th>
</tr>
</thead>
<tbody>
<tr>
<td>$S \rightarrow NP \ VP$</td>
<td>$NP \ VP$</td>
</tr>
<tr>
<td>$NP \rightarrow N$</td>
<td>$N \ VP$</td>
</tr>
<tr>
<td>$N \rightarrow John$</td>
<td>$John \ VP$</td>
</tr>
<tr>
<td>$VP \rightarrow V \ NP$</td>
<td>$John \ V \ NP$</td>
</tr>
<tr>
<td>$V \rightarrow kissed$</td>
<td>$John \ kissed \ NP$</td>
</tr>
<tr>
<td>$NP \rightarrow N$</td>
<td>$John \ kissed \ N$</td>
</tr>
<tr>
<td>$N \rightarrow Lisa$</td>
<td>$John \ kissed \ Lisa$</td>
</tr>
</tbody>
</table>
Terminology

- A grammar is a 4-tuple
  
  (non-terminals, terminals, productions, start-symbol)
  
  or

  \[(N, \Sigma, P, S)\]

- A production is of the form \( \alpha \rightarrow \beta \) where \( \alpha, \beta \) are taken from \( N \cup \Sigma \).
- Read \( \alpha \rightarrow \beta \) as “rewrite \( \alpha \) with \( \beta \)”.
- Read \( \Rightarrow \) as “directly derives”.
- Read \( \Rightarrow^r \) as “directly derives using rule \( r \)”.
- Read \( \Rightarrow^* \) as “derives in one or more steps”.
A Simple PL Grammar

Here’s a grammar for a simple programming language:

\[
\begin{align*}
\text{Program} &::= \text{BEGIN Stat END} \\
\text{Stat} &::= \text{ident := Expr} \\
\text{Expr} &::= \text{Expr + Expr} \mid \\
& \quad \text{Expr * Expr} \mid \\
& \quad \text{ident} \mid \text{number}
\end{align*}
\]

- We write terminal symbols like this.
- We write non-terminal symbols like this.
- Sometimes we write \( : := \) instead of \( \rightarrow \).
- \( A \rightarrow b \mid c \) is the same as \( A \rightarrow b; A \rightarrow c \). Read \( \mid \) as “or”.
A Simple PL Grammar...

We know the sentence

\[
\text{BEGIN } a := 5 + 4 \times 3 \text{ END}
\]

is in the language because we can derive it from the start symbol:

\[
\begin{align*}
\text{Program} & \Rightarrow \text{BEGIN Stat END} \\
& \Rightarrow \text{BEGIN ident := Expr END} \\
& \Rightarrow \text{BEGIN } "a" := \text{Expr END} \\
& \Rightarrow \text{BEGIN } "a" := \text{Expr + Expr END} \\
& \Rightarrow \text{BEGIN } "a" := 5 + \text{Expr END} \\
& \Rightarrow \text{BEGIN } "a" := 5 + \text{Expr} \times \text{Expr END} \\
& \Rightarrow \text{BEGIN } "a" := 5 + 4 \times \text{Expr END} \\
& \Rightarrow \text{BEGIN } "a" := 5 + 4 \times 3 \text{ END}
\end{align*}
\]
Terminology...

Our English grammar is the 4-tuple
\[
\left( \{ S, NP, V, \ldots \}, \right.
\left. \{ John, house, died, \ldots \}, \right.
\left. \{ S \rightarrow NP \ VP, VP \rightarrow V, \ldots \}, \right.
\left. S \right) \]

Our PL grammar is the 4-tuple
\[
\left( \{ Program, Stat, \ldots \}, \right.
\left. \{ BEGIN, :=, *, \ldots \}, \right.
\left. \{ Program \ := \ BEGIN \ Stat \ END, \ldots \}, \right.
\left. Program \right) \]
We often want to show how a particular sentence was derived. We can do this without listing all the steps explicitly by drawing a parse tree.

A parse tree is a tree where

1. The root is labeled by the start symbol.
2. Each leaf is labeled by a terminal symbol.
3. Each interior node is labeled by a non-terminal symbol.
If one step of our derivation is

$$\cdots A \cdots \Rightarrow \cdots X Y Z \cdots$$

(i.e., we used the rule $A \rightarrow XYZ$) then we’ll get a parse (sub-)tree
Parse Trees...

\[
S \rightarrow \text{NP VP} \\
\text{NP} \rightarrow \text{N} \\
\text{N} \rightarrow \text{John} \\
\text{VP} \rightarrow \text{V NP} \\
\text{V} \rightarrow \text{kissed} \\
\text{NP} \rightarrow \text{N} \\
\text{N} \rightarrow \text{Lisa} \\
\text{VP} \rightarrow \text{V NP} \\
\text{V} \rightarrow \text{kissed} \\
\text{NP} \rightarrow \text{N} \\
\text{N} \rightarrow \text{Lisa}
\]

\[S \]

\[
\text{NP} \rightarrow \text{John} \\
\text{V} \rightarrow \text{kissed} \\
\text{NP} \rightarrow \text{Lisa}
\]

John kissed Lisa
Parse Trees...

Program  ⇒  BEGIN  Stat  END
⇒  BEGIN  ident  :=  Expr  END
⇒  BEGIN  "a"  :=  Expr  END
⇒  BEGIN  "a"  :=  Expr  +  Expr  END
⇒  BEGIN  "a"  :=  5  +  Expr  END
⇒  BEGIN  "a"  :=  5  +  Expr  *  Expr  END
⇒  BEGIN  "a"  :=  5  +  4  *  3  END

Program
  BEGIN
  Stat
  END
ident
  :=
  Expr
"a"
Expr
        +
Expr
5
Expr
        *
Expr
4
3
A grammar is regular if all rules are of the form

\[ A \rightarrow aB \]

\[ A \rightarrow a \]

By convention, the symbols \( A, B, C, \ldots \) are non-terminals, \( a, b, c, \ldots \) are terminals, and \( \alpha, \beta, \gamma, \ldots \) are strings of symbols.

Regular grammars are used to describe the lexical structure of programs, i.e. what tokens look like.
Context-Free Grammars

- Programming language syntax is described by a context free grammar (CFG).
- In a CFG all rules are of the form

\[ A \rightarrow \gamma \]

\( \gamma \) is any sequence of terminals or non-terminals. \( A \) is a single non-terminal.

- Example: an if-statement consists of an if-token, expression, then-token, statement, and (maybe) an else-token followed by a statement.
BNF is Backus-Naur Form, a way to write CFGs. EBNF (Extended BNF) is a more expressive way to write CFGs.

Repetition and choice are common structures in a language (and hence, its grammar).

Repetition:
\[
\text{int } x, y, z, w, \ldots; \]

Choice:
\[
\text{class C } \{ \ldots \}
\]
\[
\text{class C extends D } \{ \ldots \}
\]
In BNF, our variable declaration

\[
\text{int } x, y, z, w, \ldots; \]

looks like this:

\[
\text{vars ::= ident ident idlist ;}
\]

\[
\text{idlist ::= , ident idlist | } \epsilon
\]

In EBNF, it looks like this:

\[
\text{vars ::= ident ident \{ , ident \} ;}
\]

I.e. \{e\} means that e is repeated 0 or more times.
In BNF, our class declaration

\[
\text{class C extends D \{ ... \}}
\]

looks like this:

\[
\text{class ::= class ident extends \{ ... \}}
\]

\[
\text{extends ::= extends ident | } \epsilon
\]

In EBNF, it looks like this:

\[
\text{class ::= class ident [extends ident] \{ ... \}}
\]

I.e. \([e]\) means that \(e\) is optional.
EBNF for Luca

program ::= 
  PROGRAM ident ; decl_list block .

decl_list ::= 
  { declaration ; } 

declaration ::= 
  VAR ident : ident | 
  TYPE ident = RECORD [ [field_list] ] | 
  TYPE ident = ARRAY expression OF ident | 
  CONST ident : ident = expression | 
  PROCEDURE ident ( [formal_list] ) decl_list block ;
EBNF for Luca...

field_list ::= field_decl {; field_decl }
field_decl ::= ident : ident
formal_list ::= formal_param {; formal_param }
formal_param ::= [VAR] ident : ident
actual_list ::= expression {, expression }
block ::= BEGIN stat_seq END
stat_seq ::= { statement; }
EBNF for Luca...

\[
\text{statement} ::= \text{designator} ::= \text{expression} \mid \text{WRITE } \text{expression} \mid \text{READ } \text{designator} \mid \text{WRITELN } \text{id}ent \( \text{[ } \text{actual} \text{list } \text{]} \) \mid \text{IF } \text{expression} \text{ THEN } \text{stat} \_ \text{seq} \text{[ } \text{ELSE } \text{stat} \_ \text{seq} \text{]} \text{ ENDIF } \mid \text{FOR } \text{ident} ::= \text{expression} \text{ TO } \text{expression} \text{[ } \text{BY } \text{expression} \text{]} \text{ DO } \text{stat} \_ \text{seq} \text{ ENDFOR } \mid \text{WHILE } \text{expression} \text{ DO } \text{stat} \_ \text{seq} \text{ ENDDO } \mid \text{REPEAT } \text{stat} \_ \text{seq} \text{ UNTIL } \text{expression} \mid \text{LOOP } \text{stat} \_ \text{seq} \text{ ENDLOOP } \mid \text{EXIT}\]

EBNF for Luca...

expression ::= 
expression bin_operator expression | unary_operator expression | 
( expression ) | 
real_literal | integer_literal | char_literal | string_literal | 
designator |

designator ::= 
ident | designator [ expression ] | designator ::= ident |

unary_operator ::= _ | TRUNC | FLOAT | NOT |

bin_operator ::= _ | * | / | % | <= | <= | == | # | >= | > | AND | OR
Ambiguous Grammars

A grammar is ambiguous if some string of tokens can produce two (or more) different parse trees.

\[ E ::= E + E \mid E \times E \mid \text{number} \]

\[
\begin{align*}
E &\Rightarrow E + E \\
&\Rightarrow 5 + E \\
&\Rightarrow 5 + E \times E \\
&\Rightarrow 5 + 4 \times E \\
&\Rightarrow 5 + 4 \times 3
\end{align*}
\]

\[
\begin{align*}
E &\Rightarrow E \times E \\
&\Rightarrow E \times 3 \\
&\Rightarrow E + E \times 3 \\
&\Rightarrow 5 + 4 \times 3
\end{align*}
\]
Ambiguities occur in natural languages also:

\[ \text{I saw the man with binoculars} \]

\[ \text{I saw the man with binoculars} \]
Operator Precedence

The precedence of an operator is a measure of its binding power, i.e. how strongly it attracts its operands.

Usually $\times$ has higher precedence than $+$:

$$4 + 5 \times 3$$

means

$$4 + (5 \times 3),$$

not

$$(4 + 5) \times 3.$$

We say that $\times$ binds harder than $\pm$. 
The **associativity** of an operator describes how operators of equal precedence are grouped.

+ and − are usually **left associative**:

\[
4 - 2 + 3
\]

means

\[
(4 - 2) + 3 = 5,
\]

not

\[
4 - (2 + 3) = -1.
\]

We say that + **associates to the left**.

^ associates to the right:

\[
2^3^4 = 2^{(3^4)}.
\]
## Operators in C

<table>
<thead>
<tr>
<th>OPERATOR</th>
<th>KIND</th>
<th>PREC</th>
<th>ASSOC</th>
</tr>
</thead>
<tbody>
<tr>
<td>a[k]</td>
<td>Primary</td>
<td>16</td>
<td></td>
</tr>
<tr>
<td>f(…)</td>
<td>Primary</td>
<td>16</td>
<td></td>
</tr>
<tr>
<td>.</td>
<td>Primary</td>
<td>16</td>
<td></td>
</tr>
<tr>
<td>-&gt;</td>
<td>Primary</td>
<td>16</td>
<td></td>
</tr>
<tr>
<td>a++, a--</td>
<td>Postfix</td>
<td>15</td>
<td></td>
</tr>
<tr>
<td>++a, --a</td>
<td>Unary</td>
<td>14</td>
<td></td>
</tr>
<tr>
<td>~</td>
<td>Unary</td>
<td>14</td>
<td></td>
</tr>
<tr>
<td>!</td>
<td>Unary</td>
<td>14</td>
<td></td>
</tr>
<tr>
<td>-</td>
<td>Unary</td>
<td>14</td>
<td></td>
</tr>
<tr>
<td>&amp;</td>
<td>Unary</td>
<td>14</td>
<td></td>
</tr>
<tr>
<td>*</td>
<td>Unary</td>
<td>14</td>
<td></td>
</tr>
<tr>
<td>*, /, %</td>
<td>Binary</td>
<td>13</td>
<td>Left</td>
</tr>
<tr>
<td>+, -</td>
<td>Binary</td>
<td>12</td>
<td>Left</td>
</tr>
<tr>
<td>&lt;&lt;, &gt;&gt;</td>
<td>Binary</td>
<td>11</td>
<td>Left</td>
</tr>
<tr>
<td>&lt;, &gt;, &lt;=, &gt;=</td>
<td>Binary</td>
<td>10</td>
<td>Left</td>
</tr>
<tr>
<td>== !=</td>
<td>Binary</td>
<td>9</td>
<td>Left</td>
</tr>
<tr>
<td>&amp; !</td>
<td>Binary</td>
<td>8</td>
<td>Left</td>
</tr>
<tr>
<td>^</td>
<td>Binary</td>
<td>7</td>
<td>Left</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>Binary</td>
</tr>
<tr>
<td>&amp; &amp;</td>
<td>Binary</td>
<td>5</td>
<td>Left</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>Binary</td>
</tr>
<tr>
<td>? :</td>
<td>Ternary</td>
<td>3</td>
<td>Right</td>
</tr>
<tr>
<td>=, +=, -=, *=, /=, %=, &lt;&lt;=, &gt;&gt;=, &amp;=, ^=,</td>
<td>=</td>
<td>Binary</td>
<td>2</td>
</tr>
</tbody>
</table>

### Table of Operators

<table>
<thead>
<tr>
<th>OPERATOR</th>
<th>KIND</th>
<th>PREC</th>
<th>ASSOC</th>
</tr>
</thead>
<tbody>
<tr>
<td>+=, -=, *=, /=, %=, &lt;&lt;=, &gt;&gt;=, &amp;=, ^=,</td>
<td>=</td>
<td>Binary</td>
<td>1</td>
</tr>
</tbody>
</table>
Expression Grammars

- We must write unambiguous expression grammars that reflect the associativity and precedence of all operators.
- The next slide gives the algorithm for writing such grammars.

Resulting Expression Grammar:

```plaintext
expr ::= expr + term | term
term ::= term * factor | factor
factor ::= ( expr ) | number
```
1. Create one non-terminal for each precedence level, for example $p_1, p_2, \cdots, p_n$, where $p_n$ has the highest precedence level.

2. For operator $\text{op}$ at precedence level $i$ construct the following production if the operator is
   ● left associative:
   $$p_i ::= p_i \ \text{op} \ p_{i+1} \ | \ p_{i+1}$$
   ● right associative:
   $$p_i ::= p_{i+1} \ \text{op} \ p_i \ | \ p_{i+1}$$

3. Construct a production for nonterminal $p_{n+1}$ which represents primary expressions such as identifiers, numbers, parenthesized expressions, etc:
   $$p_{n+1} ::= (p_1) \ | \ \text{num} \ | \ \text{id}$$
Expression Grammars...

\[ E ::= E + T \mid T \]
\[ T ::= T * F \mid F \]
\[ F ::= \text{number} \]

\[
E \Rightarrow E + T \\
\Rightarrow T + T \\
\Rightarrow F + T \\
\Rightarrow 5 + T \\
\Rightarrow 5 + T * F \\
\Rightarrow 5 + F * F \\
\Rightarrow 5 + 4 * F \\
\Rightarrow 5 + 4 * 3
\]

\[
E \Rightarrow E + T \\
\Rightarrow E + T * F \\
\Rightarrow E + T * 3 \\
\Rightarrow E + F * 3 \\
\Rightarrow E + 4 * 3 \\
\Rightarrow T + 4 * 3 \\
\Rightarrow F + 4 * 3 \\
\Rightarrow 5 + 4 * 3 \\
\Rightarrow 5 + 4 * 3
\]
Abstract Syntax

- We distinguish between a language’s **concrete** and **abstract** syntax.
- The concrete syntax describes the textual layout of programs written in the language, eg. what **if**-statements look like.
- The abstract syntax describes the **logical** structure of the language; eg. that **if**-statements consist of three parts (expression, statement, statement).
Abstract Syntax...

- The abstract syntax also describes the structure of the abstract syntax tree (AST).
- Each abstract syntax rule represents the structure of an AST node-type.
- A parser converts from the program’s concrete syntax to its corresponding abstract syntax, i.e. it reads the source code of the input program and produces an AST.
Grammar Example I

Concrete Grammar:

\[ S ::= \text{ident} ::= E \mid \text{if} \ E \ \text{then} \ SS_1[\text{else} \ SS_2] \ \text{end} \mid \text{while} \ E \ \text{do} \ SS \ \text{end} \mid \epsilon \]

\[ SS ::= S ; SS \mid \epsilon \]

Abstract Grammar:

\[ \text{Assign} ::= \text{ident} \ \text{Expr} \]

\[ \text{If} ::= \text{Expr} \ \text{StatSeq} \]

\[ \text{IfElse} ::= \text{Expr} \ \text{StatSeq} \ \text{StatSeq} \]

\[ \text{While} ::= \text{Expr} \ \text{StatSeq} \]

\[ \text{Stat} ::= \text{Assign} \mid \text{If} \mid \text{IfElse} \mid \text{While} \]

\[ \text{StatSeq} ::= \text{Stat} \ \text{StatSeq} \mid \text{NULL} \]
The rule

\[
\text{IfElse ::= Expr StatSeq StatSeq}
\]

says that an if-statement consists of three parts, or, equivalently, that an AST if-node will have three children:

```
+-----+-----+-----+
|     | Expr | Stat |
|     +-----+-----+
|       Stat | Stat |
|       +-----+-----+
|       +-----+-----+
| IfElse +-----+-----+
```

We use recursive rules to define lists (e.g. declaration-lists, statement-lists):

\[
\text{StatSeq ::= Stat StatSeq | NULL}
\]
Grammar Example I...

Stat ::= Assign | If | IfElse | While
StatSeq ::= Stat StatSeq | NULL

if a then
while x do
  c := 5;
  d := 6;
end;
end;

[37]
Concrete Grammar Example II

Program ::= program ident ; DeclSeq begin StatSeq end .
DeclSeq ::= Decl ; DeclSeq | ε
Decl ::= var ident : ident
Stat ::= ident ::= Expr | if Expr then StatSeq else StatSeq
StatSeq ::= Stat ; StatSeq | ε
Expr ::= ident | const

Example:

PROGRAM P;
  VAR I : INTEGER;
  VAR C : CHAR;
  VAR J : INTEGER;
BEGIN I := 6; J := I; END.
Abstract Grammar...

Some items in the grammar are **attributes** (names of identifiers, e.g.) some are **children** (expression & statements in an if-statement, e.g.).

Every child & attribute in the abstract grammar is given a name:

\[ \text{LOP} : \text{Expr}. \]

Example:

\[ \text{IfStat} ::= \text{Expr} : \text{Expr} \text{ Then} : \text{Stat} \text{ Else} : \text{Stat} \]
Abstract Grammar...

- **Input attributes** are data (e.g. identifiers, constants) created by the lexer/parser. I write them:

  \[ \text{Name} : \text{String}. \]

- Example:

  \[ \text{IntConst} ::= \text{Value} : \text{INTEGER} \leftrightarrow \text{Pos} : \text{Position} \]

- I prefer linked lists to recursion to define lists. A statement sequence are statements linked on a child \text{Next} : \text{StatSeq}. Lists end with an empty node: \text{NoDecl}. 
Grammar Example...

Abstract Grammar:

```
Program ::= Name:String DeclSeq:Decl
    StatSeq:StatSeq Pos:Position
Decl ::= VarDecl | ProcDecl | ··· | NoDecl
VarDecl ::= Name:String TypeName:String Pos:Position
    Next:Decl
Stat ::= Assign | IfStat | ··· | NoStat
Assign ::= Des:Name Expr:Expr Pos:Position Next:Stat
    Next:Stat
Expr ::= Name | IntConst
Name ::= Name:String Pos:Position
IntConst ::= Value:INTEGER Pos:Position
```

520—Spring 2005—51
PROGRAM P;
VAR I : INTEGER;
VAR J : INTEGER;
VAR C : CHAR;
BEGIN
I := 6;
J := I;
END.
Grammar Example III

Assign ::= \texttt{ident} ::= Expr

Expr ::= Expr + Term | Term

Term ::= Term * Factor | Factor

Factor ::= ( Expr ) | \texttt{ident} | \texttt{const}

---

Abstract Grammar (A):

Assign ::= \texttt{Des:Name Expr:Expr} \iff \texttt{Pos:Position}

Expr ::= BinOp | Name | IntConst

BinOp ::= \texttt{LOP:Expr} \iff \texttt{Op:}(Add,Mul) \texttt{ROP:Expr} \iff \texttt{Pos:Position}

Name ::= \iff \texttt{Name:String} \iff \texttt{Pos:Position}

IntConst ::= \iff \texttt{Value:INTEGER} \iff \texttt{Pos:Position}
There is often more than way to design the abstract grammar.

We can turn attributes into node-kinds and vice versa.

**Abstract Grammar (B):**

```plaintext
Assign ::= Des:Name Expr:Expr ⇐ Pos:Position
Expr ::= Add | Mul | Name | IntConst
Add ::= LOP:Expr ROP:Expr ⇐ Pos:Position
Mul ::= LOP:Expr ROP:Expr ⇐ Pos:Position
Name ::= ⇐ Name:String ⇐ Pos:Position
IntConst ::= ⇐ Value:INTEGER ⇐ Pos:Position
```
Grammar Example III...

\[ I := J \ast 5 \ast (K + 3) \]
Grammar Example III...

\[ I := J \times 5 \times (K + 3) \]
Compiler Grammars

**Concrete Grammar**

```
PROG ::= STATS
STATS ::= [stat ;]+
stat ::= id := expr
expr ::= expr + expr | id | int
```

**Abstract Grammar**

```
IDENT=L(L|D)*
INT=D+
FLOAT=D+.D+
L=a|...|z|A|...|Z
D=0|...|9
```

**Parser**

Tokens → Syntax Analysis → Concrete Grammar → Abstract Grammar → AST

**Lexical Grammar**

```
L::=a|...|z|A|...|Z
D::=0|...|9
```

**Lexical Grammar**

```
program(id):= PROGRAM id ; stats
stats ::= [stat ;]+
stat ::= id := expr
expr ::= expr + expr | id | int
```

**Parser**

```
PROG ::= STATS
STATS ::= ASSIGN STATS | NULL
ASSIGN ::= ID Expr
Expr ::= ID | INT | ADD
ADD ::= Expr Expr
```
# The Chomsky Hierarchy

<table>
<thead>
<tr>
<th>TYPE</th>
<th>GRAMMAR</th>
<th>PSR</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>Unrestricted</td>
<td>( \alpha \rightarrow \beta )</td>
</tr>
<tr>
<td>1</td>
<td>Context Sensitive</td>
<td>( \alpha \rightarrow \beta, )</td>
</tr>
<tr>
<td></td>
<td></td>
<td>(</td>
</tr>
<tr>
<td>2</td>
<td>Context Free</td>
<td>( A \rightarrow \beta )</td>
</tr>
<tr>
<td>3</td>
<td>Regular</td>
<td>( A \rightarrow a\beta )</td>
</tr>
<tr>
<td></td>
<td></td>
<td>( A \rightarrow a )</td>
</tr>
</tbody>
</table>
The Chomsky Hierarchy...

- Regular languages are less powerful than context free languages.
- Languages are organized in the Chomsky Hierarchy according to their generative power.
- Type 3 languages are more restrictive (can describe simpler languages than) type 2 languages.
- Type 3 languages can be parsed in linear time, type 2 languages in cubic time.
- Programming languages are in between type 2 and 3.
- Two natural languages (Swiss German and Bambara) are known not to be context free.
Noam Chomsky

Linguist, social/political theorist; born in Philadelphia. Son of a distinguished Hebrew scholar, he was educated at the University of Pennsylvania, where he was especially influenced by Zellig Harris; after taking his M.A. there in 1951, he spent four years as a junior fellow at Harvard (1951–55), then was awarded a Ph.D. from the University of Pennsylvania (1955). In 1955 he began what would be his long teaching career at the Massachusetts Institute of Technology. He became known as one of the principal founders of transformational-generative grammar, a system of linguistic analysis that challenges much traditional linguistics and has much to do with philosophy, logic, and psycholinguistics; his book Syntactic Structures (1957) was credited with revolutionizing the discipline of linguistics.
Chomsky’s theory suggests that every human utterance has two structures: surface structure, the superficial combining of words, and "deep structure," which are universal rules and mechanisms. In more practical terms, the theory argues that the means for acquiring a language is innate in all humans and is triggered as soon as an infant begins to learn the basics of a language. Outside this highly rarefied sphere, Chomsky early on began to promote his radical critique of American political, social, and economic policies, particularly of American foreign policy as effected by the Establishment and presented by the media; he was outspoken in his opposition to the Vietnam War and later to the Persian Gulf War. His extensive writings in this area include American Power and the New Mandarins (1969) and Human Rights and American Foreign Policy (1978).
Noam Chomsky...

- “If the Nuremberg laws were applied today, then every Post-War American president would have to be hanged.”
- “The corporatization of America during the past century has been an attack on democracy.”
- “Any dictator would admire the uniformity and obedience of the [U.S.] media.”
- “Judged in terms of the power, range, novelty and influence of his thought, Noam Chomsky is arguably the most important intellectual alive.” (The New York Times Book Review)
- Chomsky on terrorism:
Chomsky vs B. F. Skinner: Famous debate in the late 50’s, early 60’s. Skinner was a behaviorist, believing that children learn language by imitating their parents. Chomsky refuted this, claiming that we all have innate language mechanisms.

Nim Chimpsky was taught sign language in 1970s. It was a lost cause. He could ask for things, but not much more.
Summary

- The job of a parser is to convert from concrete syntax to abstract syntax.
- We use context free grammars to describe both the concrete and the abstract syntax.
- The concrete syntax is described in the language manual of the language we’re compiling.
- The abstract syntax we make up ourselves. There are many ways to define the abstract syntax of a language and personal preference will play a role in how we construct it.
Readings and References

- Read Scott, Chapter 2: Programming Language Syntax

- Read Louden:
  - Regular Expressions 34–47.
  - Context-Free Grammars 95–142.

- or the Dragon Book:
  - grammars 165–171
  - associativity & precedence 30–32
  - ambiguity 171, 174–175
  - derivations 167–169
  - parse trees 169–171
  - top-down parsing 41–43
  - left recursion 47–48
Use this abstract syntax to draw an AST for the TINY program below:

```
BEGIN
    INT x;
    PRINT x + 9.9;
END
```

**Abstract Syntax:**

- PROGRAM → STATSEQ
- STATSEQ → STAT STATSEQ | NULL
- STAT → ASSIGN | PRINT | DECL
- DECL → ident type
- ASSIGN → ident EXPR
- PRINT → EXPR
- EXPR → BINOP | IDENT | INTLIT
- BINOP → op EXPR EXPR
- IDENT → ident
- INTLIT → int
- FLTLIT → fbat